9 PJL File System Commands

Introduction

Several HP LaserJet printers have mass storage capabilities for storing fonts, macros, electronic forms, and other information. Printers such as the HP LaserJet 4000 and 5000 series printers have an optional disk drive. The HP LaserJet 4000, 5000 series and some other printers support flash memory, which is non-volatile memory contained in modules that are inserted into the printer's SIMM/DIMM slots. Preferably, the printer's mass memory is managed by a host-based mass storage application that supports interaction between host software and the printer disk and/or flash memory. However, users in some environments might not have a mass storage management application. The PJL file system commands described in this chapter are provided in order to make use of printer- based mass storage when a mass storage management application is not available. The PJL file system commands described in this chapter are used the same on both types of printer mass storage (printer disk drive and flash memory).

Note For information about creating a host-based mass storage management application, contact your HP support representative.

The PJL file system consists of eight commands:

- FSAPPEND
- FSDELETE
- FSDIRLIST
- FSDOWNLOAD
- FSINIT
- FSMKDIR
- FSQUERY
- FSUPLOAD

These commands are described on the following pages. For feature support information, see Appendix A.

FSAPPEND Command

The FSAPPEND command appends data to an existing file, or if the file doesn't exist, creates the file and loads it with the given data.

Syntax:

```
@PJL FSAPPEND FORMAT:BINARY SIZE=integer
↩NAME = "pathname" [<CR>]<LF>
<binary data><ESC>%-12345X
```

Parameters:

Parameter	Functional Range	Default
SIZE=*integer*	0 to $2^{31} - 1$	N/A
NAME = "*pathname*"	Roman-8 characters 01 thru 255	N/A

- **SIZE = *integer*** — The size variable indicates the number of bytes in the file to be appended. The size indicates the number of bytes immediately following the <LF> to the next UEL command.
- **NAME = "*pathname*"** — The variable *pathname* is similar to an MS-DOS filename, except it is not limited to eight characters and a three-character extension. The pathname may contain up to 100 Roman-8 characters per *item* in the range 01 through 255, but the first and last characters must not be a space character (character 32), or character 229. Each *item* is delimited by the backslash " \ " character. If more than one backslash character is used (such as \\\\ ), the file system treats it as one backslash. The maximum number of characters in a pathname is 255; the maximum number of items is 9. The volume of the PJL file system is required in the pathname (the volume range is 0, 1, and 2, depending on the printer [prior to LaserJet 4000, the only valid volume was 0] — see the examples on the following page).

Note For the FSAPPEND command, the pathname must be a file (not a directory).

Examples of valid pathnames include:

0:	Volume 0
0:\	Root directory on volume 0
0:\MyDir	*MyDir* directory
0:\My/Dir\Gen/Dir	*Gen/Dir* directory in *My/Dir* directory
0:\dir0\file1	*file1* file in *dir0* directory
0:\DIR1\DIR2\File1	*File1* file in *DIR2* directory in *DIR1* directory
0:\DIR1\Prefix/Suffix	*Prefix/Suffix* file in *DIR1* directory

- ***<binary data>*** — This is the binary file data to be appended or used to create a new file.

Example: Using the FSAPPEND Command

This example demonstrates using the FSAPPEND command to append a file.

```
@PJL FSAPPEND FORMAT:BINARY
➥NAME = "0:\pcl\macros\OurLogo"
➥SIZE = 35<CR><LF>
35 bytes of macro data<ESC>%-12345X
```

Related Commands:

FSDELETE, FSDIRLIST, FSDOWNLOAD, FSINIT, FSMKDIR, FSQUERY, FSUPLOAD

FSDIRLIST Command

The FSDIRLIST command returns a list of files and directories which exist within the specified directory on the printer's file system. This command is similar in function to the DOS DIR command. The ENTRY and COUNT parameters are used to limit the amount of data returned to the host.

Syntax:

```
@PJL FSDIRLIST NAME = "pathname"
↪ENTRY=integer COUNT=integer[<CR>]<LF>
```

Response Syntax:

```
@PJL FSDIRLIST NAME = "pathname"
↪ENTRY=integer[<CR>]<LF>
filename TYPE=FILE SIZE=integer<CR><LF>
filename TYPE=DIR<CR><LF>
...
<FF>
```

If the *pathname* is invalid:

```
@PJL FSDIRLIST NAME = "pathname"[<CR>]<LF>
FILEERROR=number<CR><LF>
<FF>
```

Parameters:

Parameter	Functional Range	Default
NAME="*pathname*"	Roman-8 characters 01 thru 255	N/A
ENTRY=*integer*	1 to $2^{31} - 1$	N/A
COUNT=*integer*	1 to $2^{31} - 1$	N/A

- **NAME = "*pathname*"** — for the FSDIRLIST command, *pathname* must be a directory (not a file). For a complete description of the NAME parameter, see the *NAME = "pathname"* description on pages 9-2 and 9-3.
- **ENTRY = *integer*** — Every file or subdirectory that has been created on the file system is considered an entry. The entry variable is the number of the starting entry to be displayed. For

example, to display the directory listing beginning with line 5, you would use ENTRY=5. The entry value must be greater than 0. Also, if the entry value is larger than the actual number of entries, the response will not include any entries.

- **COUNT = *integer*** — The count variable specifies the number of entries to be returned. If the specified count is greater than the actual number of entries in the specified range, the actual number of entries will be displayed.

Printer Responses

The printer response indicates the number of bytes in the listed files (SIZE=*number of bytes*). If the pathname is invalid, a FILEERROR number is returned. This error number is equivalent to the PJL File System error numbers (32xxx) listed in Appendix D, minus the 32 and the leading zeros. For example, if the printer returns FILEERROR=17, the error is equivalent to status code 32017 (Invalid parameter), meaning the FSDIRLIST command contained an invalid parameter. See *PJL File System Errors (32xxx)* in Appendix D for a complete list of error codes.

Example: Using the FSDIRLIST Command

This example demonstrates using the FSDIRLIST command to return a directory list from the printer's file system.

```
@PJL FSDIRLIST NAME = "0:\pcl\macros" ENTRY=1
➥COUNT=25 <CR><LF>
```

A sample response would appear similar to this:

```
@PJL FSDIRLIST NAME = "0:\pcl\macros"
➥ENTRY=1<CR><LF>
. TYPE=DIR<CR><LF>
.. TYPE=DIR<CR><LF>
invoice.prn.rl  TYPE=FILE SIZE=1619<CR><LF>
page1.prn.tf    TYPE=FILE SIZE=2260<CR><LF>
gen_cond.prn.mt TYPE=FILE SIZE=900<CR><LF>
pclResourceFile TYPE=FILE SIZE=420<CR><LF>
```

Related Commands:

FSAPPEND, FSDELETE, FSDOWNLOAD, FSINIT, FSMKDIR, FSQUERY, FSUPLOAD, INFO FILESYS

FSDELETE Command

The FSDELETE command is used to delete files or empty directories from the printer's file system.

Syntax:

```
@PJL FSDELETE NAME = "pathname" [<CR>]<LF>
```

Parameters:

Parameter	Functional Range	Default
NAME = "*pathname*"	Roman-8 characters 01 thru 255	N/A

- **NAME = "*pathname*"** — For a complete description of the NAME parameter, see the *NAME = "pathname"* description on pages 9-2 and 9-3.

Example: Using the FSDELETE Command

This example demonstrates using the FSDELETE command to delete a file from the printer's file system.

```
@PJL FSDELETE NAME = "0:\pcl\macros\Name: HP
↪Logo; Application; ABC; Manufacturer: XYZ;
↪Version: 7.9; Date: 8/17/94" <CR><LF>
```

Related Commands:

FSAPPEND, FSDIRLIST, FSDOWNLOAD, FSINIT, FSMKDIR, FSQUERY, FSUPLOAD, INFO FILESYS

FSDOWNLOAD Command

The FSDOWNLOAD command downloads a file to the printer file system. If a file with the same name exists, the downloaded file overwrites it.

Syntax:

```
@PJL FSDOWNLOAD FORMAT:BINARY SIZE=int
➥NAME = "pathname" [<CR>]<LF>
<binary data><ESC>%-12345X
```

Parameters:

Parameter	Functional Range	Default
SIZE=*integer*	0 to $2^{31} - 1$	N/A
NAME = "*pathname*"	Roman-8 characters 01 thru 255	N/A

- **SIZE = *integer*** — The size variable indicates the number of bytes in the file to be downloaded. The size indicates the number of bytes immediately following the <LF> to the next UEL command.
- **NAME = "*pathname*"** — For the FSDOWNLOAD command, "*pathname*" must be a file (not a directory). For a complete description of the NAME parameter, see the *NAME = "pathname"* description on pages 9-2 – 9-3.
- ***<binary data>*** — This is the binary file data to be downloaded, such as font files and macro data.

Example: Using the FSDOWNLOAD Command

This example demonstrates using the FSDOWNLOAD command to download a file to the printer file system.

```
@PJL FSDOWNLOAD FORMAT:BINARY
➥NAME = "0:\pcl\macros\Name: HP Logo;
➥Application; ABC; Manufacturer: XYZ;
➥Version: 7.9; Date:8/7/94" SIZE = 22<CR><LF>
22 bytes of macro data<ESC>%-12345X
```

Related Commands:

FSAPPEND, FSDELETE, FSDIRLIST, FSINIT, FSMKDIR, FSQUERY, FSUPLOAD, INFO FILESYS

FSINIT Command

The FSINIT command is used to initialize the printer's mass storage file system. This FSINIT command must be part of a secure job.

Syntax:

`@PJL FSINIT VOLUME = "`*`pathname`*`" [<CR>]<LF>`

Parameters:

Parameter	Functional Range	Default
VOLUME = "*pathname*"	0:, 1:*, 2:*	N/A

* Volumes 1: and 2: are only supported on the HP LaserJet 4000 and newer printers.

- **VOLUME** = "***pathname*"** — The variable *pathname* must be a volume.

Example: Using the FSINIT Command

This example demonstrates using the FSINIT command to initialize volume 0 of the printer file system.

```
@PJL FSINIT VOLUME = "0:" [<CR>]<LF>
```

Related Commands:

FSAPPEND, FSDELETE, FSDIRLIST, FSDOWNLOAD, FSMKDIR, FSQUERY, FSUPLOAD, INFO FILESYS

FSMKDIR Command

The FSMKDIR command creates the specified directory on the printer file system.

Syntax:

```
@PJL FSMKDIR NAME = "pathname" [<CR>]<LF>
```

Parameters:

Parameter	Functional Range	Default
NAME = "*pathname*"	Roman-8 characters 01 thru 255	N/A

- **NAME = "*pathname*"** — For the FSMKDIR command, "*pathname*" must be a file (not a directory). For a complete description of the NAME parameter, see the *NAME = "pathname"* description on pages 9-2 and 9-3.

Example: Using the FSMKDIR Command

The following two command lines demonstrate using the FSMKDIR command to create directories on the printer file system.

```
@PJL FSMKDIR NAME = "0:\pcl" [<CR>]<LF>
@PJL FSMKDIR NAME = "0:\pcl\macros" [<CR>]<LF>
```

Related Commands:

FSAPPEND, FSDELETE, FSDIRLIST, FSDOWNLOAD, FSINIT, FSQUERY, FSUPLOAD, INFO FILESYS

FSQUERY Command

The FSQUERY command is used to determine if a given entry exists within the file system, the type of entry (file or directory), and if it is a file, the file size in bytes.

Syntax:

```
@PJL FSQUERY NAME = "pathname" [<CR>]<LF>
```

Response Syntax:

If *pathname* is a file:

```
@PJL FSQUERY NAME="pathname" TYPE=FILE
↪SIZE=integer<CR><LF>
<FF>
```

If *pathname* is a directory:

```
@PJL FSQUERY NAME="pathname" TYPE=DIR<CR><LF>
<FF>
```

If the *pathname* is invalid:

```
@PJL FSQUERY NAME="pathname"<CR><LF>
↪FILEERROR=number<CR><LF>
<FF>
```

Parameters:

Parameter	Functional Range	Default
NAME="*pathname*"	Roman-8 characters 01 thru 255	N/A

- **NAME = "*pathname*"** — For a complete description of the NAME parameter, see the *NAME = "pathname"* description on pages 9-2 and 9-3.

Printer Responses

The printer response indicates the number of bytes in the queried files. If the pathname is invalid, a FILEERROR number is returned. The returned number is equivalent to the PJL File System errors (32xxx) listed in Appendix D, only without the 32 and the leading zeros. For example, if the printer returns FILEERROR=7, the error is equivalent to status code 32007 (Illegal name), meaning the FSQUERY command contained an illegal NAME variable. See *PJL File System Errors (32xxx)* in Appendix D for a complete list of error codes.

Example: Using the FSQUERY Command

The following two command lines demonstrate using the FSQUERY command to check for the existence of a file and a directory on the printer's file system.

```
@PJL FSQUERY NAME = "0:\pcl\file1" <CR><LF>
@PJL FSQUERY NAME = "0:\pcl\macro\" <CR><LF>
```

Related Commands:

FSAPPEND, FSDELETE, FSDIRLIST, FSDOWNLOAD, FSINIT, FSMKDIR, FSUPLOAD, INFO FILESYS

FSUPLOAD Command

The FSUPLOAD command uploads a file, or a part of a file, from the printer file system to the host.

Syntax:

```
@PJL FSUPLOAD NAME = "pathname"
↪OFFSET=<number> SIZE=<number>[<CR>]<LF>
```

Response Syntax:

If the *pathname* is valid:

```
@PJL FSUPLOAD FORMAT:BINARY
↪NAME = "pathname" OFFSET=<number>
↪SIZE=<number><CR><LF>
<SIZE bytes of file data>
<FF>
```

If the *pathname* is invalid:

```
@PJL FSUPLOAD NAME = "pathname"<CR><LF>
FILEERROR=<number><CR><LF>
<FF>
```

Parameters:

Parameter	Functional Range	Default
NAME="*pathname*"	Roman-8 characters 01 thru 255	N/A
OFFSET=*integer*	0 to $2^{31} - 1$	N/A
SIZE=*integer*	0 to $2^{31} - 1$	N/A

- **NAME = "*pathname*"** — For the FSUPLOAD command, the variable *pathname* must be a file name (not a directory). For a complete description of the NAME parameter, see *NAME = "pathname"* on pages 9-2 to 9-3.
- **OFFSET = *integer*** — The offset variable specifies the offset, from the beginning of the file, indicating the point to begin uploading.

- **SIZE = *integer*** — The size variable indicates the number of bytes to be uploaded. If the amount of data in the file after OFFSET is less than the specified size value, the remainder of the file is returned and the SIZE value in the response indicates the actual amount of uploaded data.

Printer Responses

If the pathname is invalid the printer response includes an error number. The returned number is equivalent to the PJL File System errors (32xxx) listed in Appendix D, only without the 32 and the leading zeros. For example, if the printer returns FILEERROR=7, the error is equivalent to status code 32007 (Illegal name), meaning the FSUPLOAD command contained an illegal NAME variable. See *PJL File System Errors (32xxx)* in Appendix D for a complete list of error codes.

Example: Using the FSUPLOAD Command

This example demonstrates using the FSUPLOAD command to upload 512 bytes from a printer file, beginning with byte 25.

```
@PJL FSUPLOAD NAME = "0:\pcl\filenumber5"
↪OFFSET=25 SIZE = 512<CR><LF>
```

Related Commands:

FSAPPEND, FSDELETE, FSDIRLIST, FSDOWNLOAD, FSINIT, FSMKDIR, FSQUERY, INFO FILESYS

File System Example

The following is a sequence of commands that can be sent to any printer with flash memory or a hard disk. The example downloads and calls a macro from the flash memory or hard disk. See the sections following this example for a description of each command.

```
<ESC>%-12345X
↩@PJL FSMKDIR NAME ="0:\pcl" <CR><LF>
@PJL FSMKDIR NAME ="0:\pcl\macros" <CR><LF>
@PJL FSDOWNLOAD FORMAT:BINARY
↩NAME ="0:\pcl\macros\a_macro" SIZE=29<CR><LF>
<ESC>*p900x1500YThis is the macro
↩<ESC>%-12345X
↩<ESC>%-12345X@PJL ENTER LANGUAGE=PCL <CR><LF>
<ESC>&f1Y
<ESC>&n8W♣a_macro
<ESC>&f4X
<ESC>E
<ESC>%-12345X
↩@PJL DEFAULT DISKLOCK = ON <CR><LF>
```

Note The ↩ symbol indicates that this text string is part of the preceding line. That is, there should be no carriage return or line feed control codes at the end of the preceding line.

First Command

```
<ESC>%-12345X
```

This PJL command causes the printer to exit the active printer language and give control to PJL.

Second Command

```
↩@PJL FSMKDIR NAME ="0:\pcl" <CR><LF>
```

This PJL command creates the *pcl* directory on volume 0: of the printer's file system.

Third Command

```
@PJL FSMKDIR NAME ="0:\pcl\macros" <CR><LF>
```

This PJL command creates the *macros* sub-directory under the *pcl* directory.

Fourth Command

```
@PJL FSDOWNLOAD FORMAT:BINARY
↩NAME ="0:\pcl\macros\a_macro" SIZE=29<CR><LF>
<ESC>*p900x1500YThis is the macro
↩<ESC>%-12345X
```

This PJL command does several things. First, it specifies the download format to be binary (`FORMAT:BINARY`). Second, it specifies the macro's pathname on the file system (`NAME = "0:\pcl\macros\a_macro"`). Third, it specifies the number of bytes to be downloaded for this macro file (`SIZE=29`). Fourth, it contains the binary data to be downloaded (`<ESC>*p900x1500YThis is the macro`).

Fifth Command

```
↩<ESC>%-12345X@PJL ENTER LANGUAGE=PCL <CR><LF>
```

This command explicitly enters the PCL language.

Sixth Command

```
<ESC>&f1Y
```

The Macro ID command sets the current macro ID to 1.

Seventh Command

```
<ESC>&n8W♣a_macro
```

The Alphanumeric ID command associates the current macro ID to the string ID which is *a_macro.* In this example, the string name must be the same filename used with the FSDOWNLOAD command in the fourth command sequence.

The Alphanumeric ID command needs to know the number of bytes being sent after the terminating W. In this example, we are sending 8 bytes. The first byte, which is the ♣ symbol, is the control-byte (The ♣ symbol is the character with a decimal value of 005). This control byte indicates that the Alphanumeric ID command will be used to associate the current macro ID to the string ID. The other 7 bytes are the characters that make up the string ID (a_macro).

Eighth Command

<ESC>&f4X

The Macro Control command with a value of 4 enables automatic overlay for the macro with the last specified ID. That is, this command prints the macro.

Note that <ESC>&f2X and <ESC>&f3X will also work here.

Ninth Command

<ESC>E

The Printer Reset command causes the printer to print and eject the page. It also restores the User Default Environment and deletes temporary fonts and macros.

Tenth Command

<ESC>%-12345X

This PJL command causes the printer to exit the active printer language and return control to PJL.

Eleventh Command

@PJL DEFAULT DISKLOCK = ON <CR><LF>

This command sets the disk or flash memory to read-only. In this state, it is not possible to format the disk, download fonts or macros, or delete fonts or macros.

Example Summary

The FSDOWNLOAD command downloads data to printers which support this command. This command allows data to be downloaded to either the printers internal disk (such as available in HP LaserJet 4000 series, 5Si, 5SiMx, and 5Si Mopier printers) or flash memory (such as available in HP LaserJet 5, 5M, and 4000 series printers).

To identify what is stored on the disk or in flash memory, you can print a directory listing for that memory device. This directory listing identifies all directories, file names, and file sizes. A directory listing can be printed from the printer's control panel by selecting either the Disk Directory page or the Flash Directory page in the Test menu.

Note

A file system command sequence can be typed in a text editor and copied to the printer, but there are some places where binary data needs to be sent. This is not a problem as long as there is an ASCII character equivalent to the necessary decimal value. For example, the decimal value for Escape is 027. There is an ASCII character (←) for this decimal value, thus it can be typed in a text editor.

There might be times when a text editor will not allow you to type a character with a certain decimal value. For example, you might need to send the character whose decimal value is 000. There is no ASCII character for that value, thus a text editor will not allow you to send this byte to the printer.